Specifying and Verifying a Transformation of Recursive Functions into Tail-Recursive Functions

Axel Suárez Polo[0000−0002−8233−3751],
José de Jesús Lavalle Martínez[0000−0001−8652−3889], and
Iván Molina Rebolledo[0000−0002−2224−7026]

Benemérita Universidad Autónoma de Puebla,
Puebla, Puebla 72000, México

Abstract. It is well known that some recursive functions admit a tail recursive counterpart which have a more efficient time-complexity behavior. This paper presents a formal specification and verification of such process. A monoid is used to generate a recursive function and its tail-recursive counterpart. Also, the monoid properties are used to prove extensional equality of both functions. In order to achieve this goal, the Agda programming language and proof assistant is used to generate a parametrized module with a monoid, via dependent types. This technique is exemplified with the length, reverse, and indices functions over lists.

Keywords: Dependent Types · Formal Specification and Verification · Tail Recursion · Accumulation · Program Transformation

1 Introduction

Dependently typed programming languages provide an expressive system that allows both programming and theorem proving. Agda is an implementation of such a language [6]. Using these programming languages, it can be proven that two functions return the same output when they receive the same input, which is a property known as *extensional equality* [5].

Programs can be developed using a transformational approach, where an initial program whose correctness is easy to verify is written, and after that, it is transformed into a more efficient program that preserves the same properties and semantics [11]. Proving that the transformed program works the same way as the original program is usually done by using *algebraic reasoning* [3], but this can also be done using dependently typed programming [10], with the advantage of the proof being verified by the compiler.

The *accumulation* strategy is a well-known program transformation technique to improve the efficiency of recursive functions [4]. This technique is the focus of this paper, in which dependently typed programming is used to develop an strategy to prove extensional equality between the original recursive programs and their tail-recursive counterparts.

2 A simple example: list length

Let's start with a simple example: a function to compute the length of a list. This function can be defined recursively as follows:

```
length : List A → N
length [] = 0
length (x :: xs) = suc (length xs)
```

Nonetheless, this function requires space proportional to the length of the list. For this reason, this program can be transformed into a tail-recursive function, which can be optimized automatically by the compiler to use constant space [2]. The transformed function is shown below:

```
length-tail : List A → N → N
length-tail [] n = n
length-tail (x :: xs) n = length-tail xs (suc n)
```

In this example, it is clear to see that for any list, both functions return the same result, but this is a fact that is not encoded in Agda itself. However, we can encode this fact in Agda using dependent function types and intensional equality:

```
length≡length-tail : ∀ (xs : List A)
                    → length xs ≡ length-tail xs 0
```

As we can see, the use of dependent types allows us to make reference to the first argument in the return type of the function, and therefore, allowing us to state the fact that both functions return the same result when called with the same list. The notion of "sameness" used here is the one of *intensional equality*, which is an inductively defined family of types [7, 10] defined as follows:

```
data _≡_ {a} {A : Set a} (x : A) : A → Set a where
  instance refl : x ≡ x
```

What this means is that two terms are equal if they are exactly the same term. Additionally, in Agda, if both terms reduce to the same term, we can state that they are intensionally equal [5]. For example, `refl : 2 + 3 ≡ 5`.

Returning to the example of the list length, we need to implement a function with that signature, which we can do by pattern matching on the input list:

```
length≡length-tail : ∀ (xs : List A)
                    → length xs ≡ length-tail xs 0
length≡length-tail [] = ?
length≡length-tail (x :: xs) = ?
```

We can proceed our proof by filling each of the holes[1]. The first case is trivial, because both function calls reduce to to the same term:

[1] The `?` symbols are holes in our proof, which must be filled later to complete the proof, but are useful to write the proof incrementally.

```
          length [] = 0                (by definition)
length-tail [] 0 = 0
```

Therefore, we can fill the first hole with `refl`:

```
length≡length-tail [] = refl
```

For the inductive case, we can reduce both terms instantiated with the argument, and check what is necessary to prove. Note that this can be done automatically by querying Agda and it's particularly useful when using the Agda mode in Emacs [12]. The reductions are shown below:

```
          length (x :: xs) = suc (length xs)
length-tail (x :: xs) 0 = length-tail xs (suc 0)
                        = length-tail xs 1
```

We need to prove that `suc (length xs) ≡ length-tail xs 1`. This time, we cannot simply use `refl`, because both sides do not reduce to the same term. For this reason, we can proceed to call this function recursively with the tail of the list. This is justified because of the Curry-Howard correspondence, and the fact that we are making a proof by induction. The result of the recursive call gives us the induction hypothesis:

```
length≡length-tail (x :: xs) =
  let ind-h = length≡length-tail xs
   in ?
```

The type of `ind-h` is `length xs ≡ length-tail xs 0`. The left sides of the induction hypothesis and what we are proving are almost the same. To make them match, we can apply the *congruence* property of equality, which has the following type:

```
cong : ∀ (f : A → B) {x y} → x ≡ y → f x ≡ f y
```

Applying this function to the induction hypothesis, we get the function below:

```
length≡length-tail (x :: xs) =
  let ind-h = length≡length-tail xs
      suc-cong = cong suc ind-h
   in ?
```

The new term `suc-cong` has the type:

```
suc (length xs) ≡ suc (length-tail xs 0)
```

The proof can be completed by making use of `transitivity`, which is represented using dependent types as follows:

```
trans : ∀ {x y z} → x ≡ y → y ≡ z → x ≡ z
```

Therefore, now our proof is:

```
length≡length-tail (x :: xs) =
  let ind-h = length≡length-tail xs
      suc-cong = cong suc ind-h
   in trans suc-cong ?
```

The type of the term required to fill the hole is:

```
suc (length-tail xs 0) ≡ length-tail xs 1
```

We need to "pull" the `1` from the accumulator somehow, and convert it to a `suc` call. We can extract this new goal into a helper function:

```
length-pull : ∀ (xs : List A)
            → suc (length-tail xs 0) ≡ length-tail xs 1
```

We can try to prove this goal by straightforward induction over the list, but we reach a dead end:

```
length-pull [] = refl
length-pull (x :: xs) = ?
```

The base case is trivial, following the definitions of the function, both terms reduce to `1`. The problem is the inductive case, which reduces as follows:

```
suc (length-tail (x :: xs) 0) = suc (length-tail xs (suc 0))
                              = suc (length-tail xs 1)
      length-tail (x :: xs) 1 = length-tail xs (suc 1)
                              = length-tail xs 2
```

So, we are left with the following goal, which is very similar to the one we started with:

```
suc (length-tail xs 1) ≡ length-tail xs 2
```

We could try to prove this proposition by straightforward induction too, but that would require us to prove the proposition for the next values `2` and `3`, and so on, *ad infinitum*.

To solve this issue, we can use a *generalization* strategy to prove this inductive property [1]. The generalized property will allows us to vary the value of the accumulator in the different cases of the inductive proof, but we will need to introduce another variable for it. It is important to note that after processing

the first `n` items of the list, we will get `n + length-tail xs 0` on the left side and `length-tail xs n` on the right one. Combining the generalization strategy and this fact, we can see that the property we have to prove is:

```
length-pull-generalized :
    ∀ (xs : List A) (n p : N)
    → n + length-tail xs p ≡ length-tail xs (n + p)
```

This function can be proved by induction over the list:

```
length-pull-generalized [] n p = refl
length-pull-generalized (x :: xs) n p = ?
```

The base case is trivial, because replacing the `xs` argument with `[]`, and following a single reduction step on both sides, a common term `n + p` is reached.

The inductive case is more interesting, reducing both sides of the equation proceeds as follows:

```
  n + length-tail (x :: xs) p = n + length-tail xs (suc p)
length-tail (x :: xs) (n + p) = length-tail xs (suc (n + p))
```

We can see that we have pretty much the induction hypothesis, with the only difference being the accumulating parameter `p`. Fortunately, as we have generalized the proposition, we can pick a value for `p` when using the induction hypothesis:

```
length-pull-generalized (x :: xs) n p =
  length-pull-generalized xs n (suc p)
```

This takes us closer to the goal we want to prove. Unfortunately, we are left with the following type after performing the substitution of `p` with `suc p`:

```
n + length-tail xs (suc p) ≡ length-tail xs (n + suc p)
```

This is almost what we want, with the exception of `suc (n + p)` not being equal to `n + suc p`. However, these two terms are indeed equal, but not definitionally, because the plus function is defined by induction on the first argument, and not on the second one:

```
_+_ : Nat → Nat → Nat
zero  + m = m
suc n + m = suc (n + m)
```

Therefore, applying reduction steps does not allow Agda to deduce the equality of these two terms. Fortunately, the fact that these terms are equal can be easily proved inductively as follows:

```
+-suc : ∀ m n → m + suc n ≡ suc (m + n)
+-suc zero    n = refl
+-suc (suc m) n = cong suc (+-suc m n)
```

The remaining step is to "replace" the `suc (n + p)` term with `n + suc p`. This could be done by using a function generalizing over the accumulator, but Agda provides the `rewrite` construct, that gives us that functionality with minimal effort:

```
length-pull-generalized (x :: xs) n p
  rewrite (sym (+-suc n p))
        = length-pull-generalized xs n (suc p)
```

We make use of the *symmetric* property of equality in the rewriting step, which allows us to flip the sides of the equality:

```
sym : ∀ {x y} → x ≡ y → y ≡ x
```

With all this in place, we can finally prove the remaining goals, giving as a result the complete proof:

```
length-pull-generalized :
    ∀ (xs : List A) (n p : ℕ)
    → n + length-tail xs p ≡ length-tail xs (n + p)
length-pull-generalized [] n p = refl
length-pull-generalized (x :: xs) n p
  rewrite (sym (+-suc n p))
        = length-pull-generalized xs n (suc p)

length-pull : ∀ (xs : List A)
            → suc (length-tail xs 0) ≡ length-tail xs 1
length-pull xs = length-pull-generalized xs 1 0

length≡length-tail : ∀ (xs : List A)
                   → length xs ≡ length-tail xs 0
length≡length-tail [] = refl
length≡length-tail (x :: xs) =
  let ind-h = length≡length-tail xs
      suc-cong = cong suc ind-h
      suc-pull = length-pull xs
   in trans suc-cong suc-pull
```

3 Another example: list reverse

The list reversal function follows a similar pattern to the one we have seen before:

```
reverse : List A -> List A
reverse [] = []
reverse (x :: xs) = reverse xs ++ (x :: [])

reverse-tail : List A -> List A -> List A
```

```
reverse-tail [] ys = ys
reverse-tail (x :: xs) ys = reverse-tail xs (x :: ys)
```

It should not come as a surprise that the equality proof is very similar too:

```
reverse-pull-generalized :
  ∀ (xs ys zs : List A)
  → reverse-tail xs ys ++ zs
      ≡ reverse-tail xs (ys ++ zs)
reverse-pull-generalized [] ys zs = refl
reverse-pull-generalized (x :: xs) ys zs =
  reverse-pull-generalized xs (x :: ys) zs

reverse-pull :
  ∀ (x : A) (xs : List A)
  → reverse-tail xs [] ++ (x :: [])
      ≡ reverse-tail xs (x :: [])
reverse-pull x xs =
  reverse-pull-generalized xs [] (x :: [])

reverse≡reverse-tail : ∀ (xs : List A)
                     → reverse xs ≡ reverse-tail xs []
reverse≡reverse-tail [] = refl
reverse≡reverse-tail (x :: xs) =
  let ind-h = reverse≡reverse-tail xs
      append-cong = cong (_++ (x :: [])) ind-h
      append-pull = reverse-pull x xs
   in trans append-cong append-pull
```

There are minor variations in the function signatures and the order of the parameters, but the structure is identical:

- Start proving by induction on the list.
- Fill the base case with `refl`.
- Take the induction hypothesis by using a recursive call.
- Apply *an operator* to both sides of the equality, using `cong`.
- Create a function to pull the accumulator, and prove it using a generalized version of this function that allows varying the accumulator.
- Compose the two equalities using the `trans` function.

4 Generalization

Starting from the function definitions, we can see that they follow the same recursive pattern, we can write this pattern in Agda, which is just an specialization of a `fold` function [8, 9]:

```
reduce : List A → R
reduce [] = empty
reduce (x :: xs) = f x <> reduce xs
```

- `R` is the result type of the function.
- `empty` is the term to return when the list is empty.
- `f` is a function to transform each element of the list into the result type.
- `<>` is the function to combine the current item and the recursive result.

In the case of the `length` function, the result type is N, the natural numbers; `empty` is `0`; the function to transform each element is a constant function that ignores its argument and returns `1`; and the function to combine the current item and the result of the recursive call is the addition function.

For the `reverse` function, the result type is the same type as the original list, `List A`; `empty` is the empty list; the function to transform each element creates just a singleton list from its parameter; and the function to combine the current transformed item and the result of the recursive call, is the flipped concatenation function. The flipping is necessary to make the function concatenate its first argument to the right:

```
reduce (x :: xs) = (λa → a :: []) x <> reduce xs
                 = (x :: []) <> reduce xs
                 = (λxs ys → ys ++ xs) (x :: []) (reduce xs)
                 = reduce xs ++ (x :: [])
```

The functions that follow this pattern, can be defined in a tail-recursive way as follows:

```
reduce-tail : List A → R → R
reduce-tail [] r = r
reduce-tail (x :: xs) r = reduce-tail xs (r <> f x)
```

We can check manually that this function matches the tail-recursive definition in the case of the `reverse` function:

```
reduce-tail (x :: xs) r = reduce-tail xs (r <> (λa → a :: []) x)
                        = reduce-tail xs (r <> (x :: [])
                        = ... xs ((λxs ys → ys ++ xs) r (x :: []))
                        = reduce-tail xs ((x :: []) ++ r)
                        = reduce-tail xs (x :: r)
```

Now we can proceed to prove that these two functions are extensionally equal in the general case. The proof follows the same pattern as the one for the `length` function:

```
reduce≡reduce-tail : ∀ (xs : List A)
                   → reduce xs ≡ reduce-tail xs empty
reduce≡reduce-tail [] = refl
reduce≡reduce-tail (x ∷ xs) =
  let ind-h = reduce≡reduce-tail xs
      op-cong = cong (f x <>_) ind-h
      op-pull = reduce-pull (f x) xs
   in trans op-cong op-pull
```

We make use of a piece of syntactic sugar called *sections*, which allows us to write the function `(λr → f x <> r)` as `(f x <>_)`. Apart from that, the proof is identical to the ones we have seen before.

However, to prove the accumulator pulling function, we need to use a different strategy. We are required to prove that:

```
reduce-pull :
  ∀ (r : R) (xs : List A)
  → r <> reduce-tail xs empty
  ≡ reduce-tail xs (empty <> r)
```

To do this, we can prove this proposition by induction over the list, which requires us to prove the proposition when `xs` is `[]`:

```
r <> reduce-tail [] empty = r <> empty
reduce-tail [] (empty <> r) = empty <> r
```

So we are required to prove that `r <> empty ≡ empty <> r`. We could require the `<>` function to be commutative, but we can "ask for less" by just requiring `empty` to be a left and right identity for `<>`, this is expressed in Agda as:

```
<>-identityˡ : ∀ (r : R) → empty <> r ≡ r
<>-identityʳ : ∀ (r : R) → r <> empty ≡ r
```

This way, we can use those identities to rewrite our goals, and make them match over the term `r`, and then, complete the base case using the trivial equality proof `refl`:

```
reduce-pull r []
  rewrite <>-identityˡ r
        | <>-identityʳ r = refl
```

The inductive case goal is:

```
r <> reduce-tail (x :: xs) empty
    = r <> reduce-tail xs (empty <> f x)
reduce-tail (x :: xs) (empty <> r)
    = reduce-tail xs ((empty <> r) <> f x)
```

Which cannot be proven directly by straightforward induction, as we have seen before, but at least we can simplify it by using the left identity property over `empty <> f x` and then over `empty <> r`:

```
reduce-pull r (x :: xs)
  rewrite <>-identity¹ (f x)
        | <>-identity¹ r
        = reduce-pull-generalized r (f x) xs
```

Finally, we just need to prove the generalized accumulation pulling function, which has the following type signature:

```
reduce-pull-generalized :
  ∀ (r s : R) (xs : List A)
  → r <> reduce-tail xs s ≡ reduce-tail xs (r <> s)
```

Note that the base case is trivial, and it is quite similar to the ones we have already proved, so we are going to focus in the inductive case. Following the same kind of reductions we have been doing before, we can see that our goal is:

```
r <> reduce-tail (x :: xs) s
    = r <> reduce-tail xs (s <> f x)
reduce-tail (x :: xs) (s <> r)
    = reduce-tail xs ((r <> s) <> f x)
```

Following the generalization strategy, we have to call the function recursively, replacing the `s` by `s <> f x`, which almost gives us what we want, except that the right hand side accumulator is associated wrongly.

```
r <> reduce-tail xs (s <> f x)
   ≡ reduce-tail xs (r <> (s <> f x))
```

Associativity is indeed the last property we need the `<>` function to satisfy. This can be expressed in Agda straightforwardly as:

```
<>-assoc : ∀ (r s t : R)
         → (r <> s) <> t ≡ r <> (s <> t)
```

Which helps us complete the proof:

```
reduce-pull-generalized r s [] = refl
reduce-pull-generalized r s (x :: xs)
  rewrite <>-assoc r s (f x)
        = reduce-pull-generalized r (s <> f x) xs
```

All of these properties match the definition of a monoid. We can complete the formalization and encapsulate it in a ready to use parametrized module, using the standard library definition of a monoid:

```
open import Algebra.Structures using (IsMonoid)

module GenericBasic
  {A : Set}
  {R : Set}
  (f : A → R)
  (empty : R)
  (_<>_ : R → R → R)
  (m : IsMonoid _≡_ _<>_ empty)
  where

open IsMonoid m using ()
  renaming ( identity^l to <>-identity^l
           ; identity^r to <>-identity^r
           ; assoc to <>-assoc
           )
```

5 Examples

6 Conclusions

Acknowledgements Please place your acknowledgments at the end of the paper, preceded by an unnumbered run-in heading (i.e. 3rd-level heading).

References

1. Abdali, S.K., Vytopil, J.: Generalization heuristics for theorems related to recursively defined functions. In: Proceedings of the Fourth AAAI Conference on Artificial Intelligence. pp. 1–5 (1984)
2. Bauer, A.: Compilation of functional programming languages using gcc—tail calls. Master's thesis, Institut für Informatik, Technische Universität München, Germany (2003)
3. Bird, R., De Moor, O.: The algebra of programming. In: NATO ASI DPD. pp. 167–203 (1996)

4. Bird, R.S.: The promotion and accumulation strategies in transformational programming. ACM Transactions on Programming Languages and Systems (TOPLAS) **6**(4), 487–504 (1984)
5. Botta, N., Brede, N., Jansson, P., Richter, T.: Extensional equality preservation and verified generic programming. Journal of Functional Programming **31** (2021)
6. Bove, A., Dybjer, P., Norell, U.: A brief overview of agda–a functional language with dependent types. In: International Conference on Theorem Proving in Higher Order Logics. pp. 73–78. Springer (2009)
7. Dybjer, P.: Inductive families. Formal aspects of computing **6**(4), 440–465 (1994)
8. Hutton, G.: A tutorial on the universality and expressiveness of fold. Journal of Functional Programming **9**(4), 355–372 (1999)
9. Meijer, E., Fokkinga, M., Paterson, R.: Functional programming with bananas, lenses, envelopes and barbed wire. In: Conference on functional programming languages and computer architecture. pp. 124–144. Springer (1991)
10. Mu, S.C., Ko, H.S., Jansson, P.: Algebra of programming using dependent types. In: International Conference on Mathematics of Program Construction. pp. 268–283. Springer (2008)
11. Pettorossi, A., Proietti, M.: Rules and strategies for program transformation. Formal Program Development pp. 263–304 (1993)
12. Wadler, P.: Programming language foundations in agda. In: Brazilian Symposium on Formal Methods. pp. 56–73. Springer (2018)